# How can i get the solution of complicated equation?

This is the equation about zeta. NSolve[eq] could not solve this equation. (I think it is too complex to solve by NSolve.)

I want to know how can I solve this equation, which has such a complicated form.

This should be the code for the equation:

eq = (0. - 30.1996 I) ζ^6 +
(0. + 9.81486 I) ζ^5 Sqrt[(0. - 640.701 I) + ζ^2] +
(0. + 20.3847 I) ζ^4 ((0. - 640.701 I) + ζ^2) +
(0. + 20.3847 I) ζ^5 Sqrt[(0. - 1127.53 I) + ζ^2] -
(0. + 20.3847 I) ζ^3 ((0. - 640.701 I) + ζ^2) Sqrt[(0. - 1127.53 I) + ζ^2] +
(0. + 9.81486 I) ζ^4 ((0. - 1127.53 I) + ζ^2) -
(0. + 9.81486 I) ζ^3 Sqrt[(0. - 640.701 I) + ζ^2] ((0. - 1127.53 I) + ζ^2) +
408.549 (
(0. - 0.630868 I) ζ^3 - (0. + 14.0909 I) ζ^2 Sqrt[(0. - 640.701 I) + ζ^2] -
(0. + 1.21667 I) ζ ((0. - 640.701 I) + ζ^2) -
(0. + 6.45738 I) ζ^2 Sqrt[(0. - 1127.53 I) + ζ^2] +
(0. + 14.7217 I) ((0. - 640.701 I) + ζ^2) Sqrt[(0. - 1127.53 I) + ζ^2] +
(0. + 0.585806 I) ζ ((0. - 1127.53 I) + ζ^2) +
(0. + 7.08825 I) Sqrt[(0. - 640.701 I) + ζ^2] ((0. - 1127.53 I) + ζ^2)) == 0;

-
You can try FindRoot; if you post actual code people will be more willing to help though. –  b.gatessucks May 24 '13 at 7:57
Also it might be helpful to know the context in which the equation came up. What problem are you trying to solve? –  Thies Heidecke May 24 '13 at 8:24
Justed added the code using OCR. –  gwr May 24 '13 at 14:28

This can be simplified quite a bit, since most of the real parts are zero:

f[ζ_] =
FullSimplify[(0. - 30.1996 I) ζ^6 + (0. + 9.81486 I) ζ^5*
Sqrt[ζ^2 - (0. + 640.701 I)] + (0. + 20.3847 I) ζ^5*
Sqrt[ζ^2 - (0. + 1127.53 I)] + (0. +
20.3847 I) ζ^4 (ζ^2 - (0. + 640.701 I)) + (0. +
9.81486 I) ζ^4 (ζ^2 - (0. + 1127.53 I)) - (0. +
9.81486 I) ζ^3*
Sqrt[ζ^2 - (0. + 640.701 I)]*(ζ^2 - (0. +
1127.53 I)) - (0. +
20.3847 I) ζ^3 (ζ^2 - (0. + 640.701 I))*
Sqrt[ζ^2 - (0. + 1127.53 I)] +
408.549 ((0. - 0.630868 I) ζ^3 - (0. + 14.0909 I) ζ^2*
Sqrt[ζ^2 - (0. + 640.701 I)] - (0. +
6.45738 I) ζ^2*
Sqrt[ζ^2 - (0. + 1127.53 I)] - (0. +
1.21667 I) ζ (ζ^2 - (0. + 640.701 I)) + (0. +
0.585806 I) ζ (ζ^2 - (0. + 1127.53 I)) + (0. +
7.08825 I)*
Sqrt[ζ^2 - (0. + 640.701 I)] (ζ^2 - (0. +
1127.53 I)) + (0. +
14.7217 I) (ζ^2 - (0. +
640.701 I)) Sqrt[ζ^2 - (0. + 1127.53 I)])]

FindRoot[f[ζ] == 0, {ζ, 7}]

(* {ζ -> 6.98478 + 0.493405 I} *)


Where my guess for the root was from a quick look at the plots:

Plot[{Re[f[ζ]], Im[f[ζ]]}, {ζ, -10, 10}]


Edit: The Rationalize[] wasn't necessary

-

## Introduction

Here is a way that finds all the solutions, all three of them. We can know this because rationalizing the equation results in a polynomial of degree 24 and NSolve finds 24 roots. The rest of the work is to select the ones that satisfy the given equation.

One should keep in mind that according to the documentation on NSolve

NSolve deals primarily with linear and polynomial equations.

## Rationalize the equation

The OP's equation is an algebraic equation that can be turned into a polynomial equation, which NSolve can then solve easily. NSolve apparently does not do this on its own, so we have to rationalize "by hand." Rationalizing the equation might introduce extraneous solutions, so we have to check the answers returned by NSolve. The somewhat nasty coefficients make it harder, since a straight rationalization of the square-roots fails: Round-off error leads to some of the square-roots not canceling out as they should. We can fix that by setting the precision of eq to Infinity.

We can find the square-roots in an expression expr with

DeleteDuplicates @ Cases[expr, Power[_, 1/2], Infinity]


Then for each square-root, we can multiply expr by the expression with the square-root replaced by its negative and simplify.

Here is the result of applying the method to function on the left hand side of the OP's eq, which is given by First @ eq:

eq2 = FullSimplify[SetPrecision[First @ eq, Infinity]];

rationalized =
Fold[
Expand @ Simplify[Times @@ (#1 /. {{}, {#2 -> -#2}})] &,
eq2,
DeleteDuplicates @ Cases[eq2, Power[_, 1/2], Infinity]];

rootsAll = NSolve[rationalized == 0, ζ]
(* 24 roots
{{ζ -> -24559.6 + 24559.6 I}, <<22>>, {ζ -> 24559.6 - 24559.7 I}}
*)


## Selecting the roots of the given equation

We use Pick roots to select a small function value, less than say 10^-1. We will check later that each is a root. A smaller threshold could be used.

rootsEq = Pick[rootsAll, Abs[First @ eq] < 10^-1 /. rootsAll]
(*
{{ζ -> -3.78042 - 5.50655 I}, {ζ -> -3.20562 + 5.39914 I}, {ζ -> 6.98478 + 0.493405 I}}
*)


### Checks

First @ eq /. rootsEq
(*
{1.49012*10^-7 + 1.19209*10^-7 I,
1.07568*10^-6 - 3.88362*10^-7 I,
3.35276*10^-7 - 1.93715*10^-7 I}
*)


These are pretty close to zero, given that the values of the derivative of the function are so great:

D[First @ eq, ζ] /. rootsEq // Abs
(*
{9.31455*10^7, 1.09303*10^8, 8.6091*10^7}
*)


We can apply FindRoot to verify and improve the NSolve solutions:

rootsEqPrime = Hold @ FindRoot[eq, {ζ, ζ0}] /. (rootsEq /. ζ -> ζ0) // ReleaseHold
(*
{{ζ -> -3.78042 - 5.50655 I}, {ζ -> -3.20562 + 5.39914 I}, {ζ -> 6.98478 + 0.493405 I}}
*)


They are virtually unchanged:

(ζ /. rootsEq) - (ζ /. rootsEqPrime)
(*
{ 0.             + 0. I,
8.88178*10^-15 - 5.32907*10^-15 I,
-4.44089*10^-15 - 1.66533*10^-15 I}
*)

-

As suggested by b.gatessucks, though this seems to produce imperfect results---and I don't understand why Intersection[out,out] does not remove repeated roots...

eq2 = FullSimplify[eq == 0];
out = ParallelTable[ ζ /. Quiet@FindRoot[eq2, { ζ, 500 RandomComplex[]}], {5000}];
test = Intersection[out, out];
tol = 10^-1;
list = {test[[1]]};
Do[
If[Norm[test[[i]] - test[[i + 1]]] > tol,
AppendTo[list, test[[i + 1]]]], {i, Length[test] - 1}];
list[[Flatten@
Position[
Chop[ eq /. ζ -> list[[#]] & /@ Range[Length[list]], 10^-6],
0]]]


This is not pretty, nor particularly robust.

-
Intersection[out, out] doesn't remove repeated roots because they are not exactly the same, only approximately the same. Use something like DeleteDuplicates[out, Chop[#1 - #2] == 0 &]. (You'll find that FindRoot[eq2, { ζ, 500 RandomComplex[]}] does a pretty bad job and produces hundreds of non-roots.) –  Michael E2 Dec 26 '13 at 16:09